DHS Data Analysis with STATA

Episode 01: Import, Clean & Explore DHS Data

Md Salek Miah
Statistician & Public Health Researcher

Dept. of Statistics, SUST   ORCID

July 2026

Course Overview

Learning Objectives

- Import and manage DHS datasets in Stata
- Interpret DHS variable naming conventions
- Define binary outcome and explanatory variables
- Handle missing data with a documented protocol
- Formally specify the complex survey design
- Produce weighted, design-consistent estimates

Dataset & Prerequisites

- **BDHS 2022**, Individual Recode (IR)
- Women aged 15–49; $n \approx 20{,}000$
- 3,000+ variables

DHS Program Data *Source: The DHS Program*

Today's Roadmap

1. **Part 1: Import & Explore** `cd, use, describe, codebook`
2. **Part 2: Understand DHS Variables** `v005, v021, v022, v024, v025`
3. **Part 3: Survey Design Theory** `weights, PSU, strata`
4. **Part 4: Define & Recode Variables** `generate, replace, recode`
5. **Part 5: Missing Data & svyset** `misstable, svyset`
6. **Part 6: Weighted Estimates** `svy: tab`

What is DHS?

The DHS Program

- USAID-funded project since 1984
- Conducted in 90+ countries
- Over 400 surveys completed
- Gold standard for population health
- Free and publicly available

Key Topics Covered

- Fertility and family planning
- Maternal and child health
- Nutrition and anthropometry
- HIV/AIDS and STIs
- Gender-based violence
- Wealth and socioeconomic status

Bangladesh DHS

BDHS conducted regularly since 1993-94, latest in 2022

Types of DHS Datasets

Individual Recode (IR)

- Women aged 15-49
- Most commonly used
- All individual-level data

Household Recode (HR)

- Household-level information
- Wealth index, assets
- Household roster data

Men's Recode (MR)

- Men aged 15-49/54
- Available in many surveys
- Limited variables

Children's Recode (KR)

- Children under 5 years
- Anthropometric measurements
- Child health indicators

Focus Today: Individual Recode (IR)

Contains all variables needed for women's health research

Importing DHS Data: Key Concepts

Before You Start

- Set your working directory – tells Stata where your files live
- Know your file type: `.dta` (Stata), `.sav` (SPSS), `.sas7bdat` (SAS)
- DHS recode files are large; confirm adequate memory before loading

Core Commands

- `cd` – change directory   `pwd` – print working directory
- `use` – load a dataset   `save` – write the current dataset to disk

Pro Tip

Always `save` under a new filename; never overwrite the original raw DHS file.

Import the DHS Dataset: Demonstration

Topics

- Opening a `.dta` file
- Setting the working directory
- Loading and saving datasets

```
cd "C:\DHS\BDHS2022"
pwd
use "BDIR81FL.dta", clear
save "bdhs2022_working.dta"
```

Exploring DHS Data: Key Concepts

Why Explore First?

- Understand the data structure
- Identify key sampling variables
- Check for missing values
- Verify data quality before modeling

Core Commands

- `describe` – names and types
- `codebook` – full variable detail
- `summarize` – basic statistics
- `tabulate` – frequency tables

Explore the Dataset: Demonstration

```
describe
codebook v012
browse
list v012 v106 in 1/10
summarize v012
tabulate v106
```

Typical DHS Dataset

$\approx$ 20,000 observations
$\approx$ 3,200 variables

Understanding Key DHS Variables

Table: Key DHS Individual Recode (IR) variables

Variable	Label	Purpose
`v005`	Sample weight	Adjusts for unequal selection probability
`v021`	PSU	Primary sampling unit (cluster)
`v022`	Strata	Sampling strata
`v024`	Region	Administrative region
`v025`	Residence	Urban / rural classification
`v012`	Age	Respondent's age (years)
`v106`	Education	Highest educational level
`v190`	Wealth	Household wealth quintile
`v130`	Religion	Religious affiliation
`v131`	Ethnicity	Ethnic group

Variable Anatomy

Name (`v005`) → **Label** ("Sample weight") → **Value label** (coded categories)

Value Labeling: Complete Examples

1. Binary/Yes-No Labels

```
* Create depression variable
gen depression = 0
replace depression = 1 if s511a == 1

* Define and attach label
label define dep_lbl 0 "No" 1 "Yes"
label values depression dep_lbl
```

2. Multiple Category Labels

```
* Education categories
label define edu_lbl 0 "No education" 1 "Primary" 2 "Secondary" 3 "Higher"
label values v106 edu_lbl

* Wealth quintiles
label define wealth_lbl 1 "Poorest" 2 "Poorer" 3 "Middle" 4 "Richer" 5 "Richest"
label values v190 wealth_lbl
```

Merging Categories: Standard Protocol

Definition

Combining two or more original categories of a variable into a single, coarser category.

When to Merge – Rules of Thumb

- Expected cell frequency < 5 in a planned crosstab (violates χ^2 assumptions)
- Categories with no theoretical distinction for the current research question
- To match a categorization used in comparable published studies
- To improve statistical power in subgroup analyses

Preferred Method: `recode` with `gen()`

`recode oldvar (range = new "label") ..., gen(newvar)` – explicit, auditable, and *non-destructive* (the original variable survives).

Merging 2 Categories: Examples

Example 1: Binary Education (Low vs High)

Original categories: No education, Primary, Secondary, Higher

```
recode v106 (0/1=0 "Low") (2/3=1 "High"), gen(edu_binary)
```

Example 2: Residence (Urban/Rural)

Original categories: 1=Urban, 2=Rural

```
recode v025 (1=1 "Urban") (2=2 "Rural"), gen(residence)
```

Example 3: Birth Size (Small vs Normal/Large)

Original categories: 1=Very small, 2=Small, 3=Average, 4=Large

```
recode m18 (1/2=1 "Small") (3/4=0 "Normal/Large"), gen(small_birth)
```

Merging 3 Categories: Examples

Example 1: Age Groups (Young, Middle, Older)

Original: Continuous age (v012)

```
recode v012 (15/24=1 "15-24") (25/34=2 "25-34") (35/49=3 "35-49"), gen(age_3cat)
```

Example 2: Education (Low, Secondary, Higher)

Original: No education, Primary, Secondary, Higher

```
recode v106 (0/1=1 "Low") (2=2 "Secondary") (3=3 "Higher"), gen(edu_3cat)
```

Example 3: Wealth (Poor, Middle, Rich)

Original: 1=Poorest, 2=Poorer, 3=Middle, 4=Richer, 5=Richest

```
recode v190 (1/2=1 "Poor") (3=2 "Middle") (4/5=3 "Rich"), gen(wealth_3cat)
```

Merging 4+ Categories: Examples

Example 1: Age Groups (7 Categories)

```
recode v012 (15/19=1 "15-19") (20/24=2 "20-24") (25/29=3 "25-29") ///
    (30/34=4 "30-34") (35/39=5 "35-39") (40/44=6 "40-44") ///
    (45/49=7 "45-49"), gen(age_7cat)
```

Example 2: Number of Children (None, 1-2, 3+)

Original: Continuous number of children

```
recode v201 (0=1 "None") (1/2=2 "1-2") (3/10=3 "3+"), gen(children_cat)
```

Example 3: Region/Province Grouping

Original: 10+ administrative regions

```
recode v024 (1/2=1 "North") (3/5=2 "Central") (6/8=3 "South") ///
    (9/10=4 "East"), gen(region_4cat)
```

Real Research Example: Lesotho DHS Study

Study: Decision-Making Autonomy and Depressive Symptoms

Based on published research: Miah & Kabir (2026), Journal of Affective Disorders Reports

Covariate Categories Used

- **Age**: 7 groups (15-19, 20-24, 25-29, 30-34, 35-39, 40-44, 45-49)
- **Age at first birth**: Early (10-17), Normal (18-24), Late (25+)
- **Age at first sex**: <18, 18+
- **Age at cohabitation**: Young (<18), Mild (18-20), Older (21+)
- **Education**: Below primary, Secondary incomplete, Secondary or higher

More Categories

- **Employment**: Currently working, Not working
- **ANC visits**: <4, 4+
- **Children**: None, 1-2, 3+
- **Birth size**: Small, Average, Large
- **Preterm birth**: Term (9+ mo), Preterm (<9 mo)
- **Wealth**: Poor, Middle, Rich
- **Residence**: Urban, Rural

Implementing Lesotho Study Categorizations

Age at First Birth

```
* v212 = age at first birth (continuous)
recode v212 (10/17=1 "Early") (18/24=2 "Normal") (25/49=3 "Late"), gen(age_first_birth)
```

Age at First Sexual Intercourse

```
* v531 = age at first intercourse
recode v531 (0/17=1 "<18 years") (18/49=2 "18+ years"), gen(age_first_sex)
```

Education Categories

```
* v106 = education
recode v106 (0=1 "Below primary") (1=2 "Primary incomplete") ///
    (2/3=3 "Secondary or higher"), gen(edu_lesotho)
```

More Lesotho Study Categorizations

Number of Children

```
* v201 = number of children ever born
recode v201 (0=1 "None") (1/2=2 "1-2") (3/20=3 "3+"), gen(children_ever)
```

Birth Size

```
* m18 = birth size (1=very small, 2=small, 3=average, 4=large)
recode m18 (1/2=1 "Small") (3=2 "Average") (4=3 "Large"), gen(birth_size)
```

Wealth Status

```
* v190 = wealth index
recode v190 (1/2=1 "Poor") (3=2 "Middle") (4/5=3 "Rich"), gen(wealth_status)
```

Creating Composite/Combined Variables

Example 1: Media Access (TV/Radio/Newspaper/Mobile)

```
* v157 = reads newspaper (0=no, 1=yes)
* v158 = listens to radio (0=no, 1=yes)
* v159 = watches TV (0=no, 1=yes)
* v170 = uses internet (0=no, 1=yes)

gen media_access = 0
replace media_access = 1 if v157==1 | v158==1 | v159==1 | v170==1
label define media_lbl 0 "No media access" 1 "Has media access"
label values media_access media_lbl
```

Example 2: IPV Justification (Combined)

```
* v744a-v744f = IPV justification variables
gen ipv_justify = 0
replace ipv_justify = 1 if v744a==1 | v744b==1 | v744c==1 | v744d==1 | v744e==1 | v744f==1
```

Advanced Labeling Techniques

Adding to Existing Labels

```
* Add new categories to existing label
label define edu_lbl 4 "Vocational", add
```

Using `label dir` and `label list`

```
* View all defined labels
label dir

* View specific label values
label list edu_lbl
```

Copying Labels to New Variables

```
* Copy value labels from original variable
clonevar edu_clean = v106
label values edu_clean edu_lbl
```

DHS Survey Design: Formal Framework

Complex Sample

DHS uses a **stratified, two-stage cluster design**: strata (`v022`) are defined by region $\times$ residence; enumeration areas serve as primary sampling units, PSUs (`v021`); households are the second-stage units.

Horvitz–Thompson Estimator

$$\hat{Y}_{HT} = \sum_{i=1}^{n} w_i \, y_i, \qquad w_i = \frac{1}{\pi_i}$$

where π_i is unit i's inclusion probability.

Consequence of Ignoring Design

- Biased point estimates
- Understated standard errors
- Invalid p-values and CIs

Resolution

```
svyset psu_var [pw=weight], strata(strata_var)
```

Sampling Weights: Construction & Interpretation

What a Weight Corrects For

- Oversampling of some strata
- Nonresponse adjustment
- Restores population representativeness

Normalization

DHS stores weights scaled by 10^6:

$$w_i = \frac{v005_i}{1{,}000{,}000}$$

```
gen wt = v005 / 1000000
```

Weighted vs. Unweighted

Unweighted statistics describe the *achieved sample*; weighted (`svy:`) statistics estimate the *target population*.

Declaring the Survey Design: `svyset`

```
gen wt = v005/1000000
svyset v021 [pw=wt], strata(v022)
```



Components

- PSU: `v021`
- Weight: `wt`
- Strata: `v022`

Why svyset?

Without `svyset`, Stata treats data as simple random sample, giving incorrect standard errors and p-values.

Handling Missing Data

Why Missing Data Matters

- Reduces sample size
- Can introduce bias
- STATA drops missing automatically

Detect and Handle

```
misstable summarize
count if missing(depression)
drop if missing(depression, age, education)
```

Best Practice

Always run `misstable summarize` before deciding on a missing-data strategy

Weighted Frequencies: Theory

Weighted Proportion

$$\hat{p} = \frac{\sum_{i=1}^{n} w_i y_i}{\sum_{i=1}^{n} w_i}$$

Implementation

```
svy: tab depression
svy: tab education
svy: tab wealth
```

Key Insight

Unweighted = Sample representation   Weighted = Population representation

Weighted Crosstabs

Purpose

- Compare outcomes across groups
- Identify disparities
- Test associations

Implementation

```
svy: tab depression education, row
svy: tab depression wealth, col
svy: tab depression residence, row
```

Interpretation

- Row %: Within row group
- Column %: Within column group
- Weighted %: Population estimates

Statistical Notes: Complex Survey Variance Estimation

Design Effect

$$\text{DEFF} = \frac{\mathrm{Var}_{\text{complex}}(\hat{\theta})}{\mathrm{Var}_{\text{SRS}}(\hat{\theta})}$$

Measures the inflation of sampling variance due to clustering and weighting

Effective Sample Size

$$n_{\text{eff}} = \frac{n}{\text{DEFF}}$$

Estimation Method

Stata's `svy:` commands default to **Taylor linearization**

Complete Analysis Workflow

Step-by-Step Process

1. Set working directory and load the dataset
2. Explore: `describe`, `codebook`, `summarize`
3. Identify key DHS design variables (`v005`, `v021`, `v022`)
4. Define the outcome variable (binary: 0/1)
5. Create explanatory variables (age groups, education, wealth)
6. Rename and label variables for clarity
7. Merge sparse categories where theoretically justified
8. Diagnose and handle missing data
9. Generate the weight and declare the survey design (`svyset`)
10. Produce weighted frequencies and crosstabs

Critical Rule

Always use `svy:` prefix for all DHS analyses

Common Mistakes to Avoid

Data Management

- Not setting working directory
- Forgetting to save datasets
- Not labeling variables
- Not documenting code

Analysis

- Forgetting survey weights
- Not using `svy:` prefix
- Ignoring missing data
- Using unweighted frequencies

How to Avoid Them

- Always start with `clear all`
- Use consistent naming conventions
- Create do-files for reproducibility
- Check codebook before analysis

Today We Learned

- What DHS is and why it matters
- Types of DHS datasets
- Importing DHS data into STATA
- Exploring variables and structure
- Understanding DHS variable codes
- DHS survey design complexity
- Defining binary outcomes
- Creating explanatory variables
- Renaming and labeling variables
- Merging 2, 3, 4+ categories
- Handling missing values
- Weighted frequencies and crosstabs

Next Episode

Univariable screening, confounder control, and DAG-based variable selection

Further Reading

- Kish, L. (1965). *Survey Sampling*. Wiley.
- ICF International (2012). *Demographic and Health Survey Sampling and Household Listing Manual*.
- Heeringa, S. G., West, B. T., & Berglund, P. A. (2017). *Applied Survey Data Analysis* (2nd ed.). CRC Press.
- Miah, M.S. & Kabir, M.B. (2026). Decision-Making Autonomy and Depressive Symptoms Among Women in Lesotho. *Journal of Affective Disorders Reports*, 25, 101080.
- StataCorp. *Survey Data Reference Manual* (current release).

Questions?

"Data analysis without survey weights is like driving without a map."

Drop your questions in the comments! Subscribe for more DHS tutorials

Contact:
saleksta@gmail.com
+8801687831074

Thank You!

Keep Learning, Keep Analyzing!

Md Salek Miah
Statistician & Public Health Researcher

Dept. of Statistics, SUST
ORCID: 0009-0005-5973-461X

saleksta@gmail.com
+8801687831074

GitHub | LinkedIn | Salek Data Lab

Turning complex data into simple public health solutions